

Variables, Constraints and Objective

Arpit Bhatia

June 5, 2019

While the last tutorial introduced you to basics of JuMP code, this tutorial will go in depth focusing on how to work with different parts of a JuMP program.

1 Variables

1.1 Variable Bounds

All of the variables we have created till now have had a bound. We can also create a free variable.

```
@variable(model, freex)
```

freex

While creating a variable, instead of using the `<=` and `>=` syntax, we can also use the `lower_bound` and `upper_bound` keyword arguments.

```
@variable(model, altx, lower_bound=1, upper_bound=2)
```

altx

We can query whether a variable has a bound using the `has_lower_bound` and `has_upper_bound` functions. The values of the bound can be obtained using the `lower_bound` and `upper_bound` functions.

```
has_upper_bound(altx)
```

```
true
```

```
upper_bound(altx)
```

```
2.0
```

Note querying the value of a bound that does not exist will result in an error.

```
lower_bound(freex)
```

```
Error: Variable freex does not have a lower bound.
```

JuMP also allows us to change the bounds on variable. We will learn this in the problem modification tutorial.

1.2 Containers

We have already seen how to add a single variable to a model using the `@variable` macro. Let's now look at more ways to add variables to a JuMP model. JuMP provides data structures for adding collections of variables to a model. These data structures are referred to as Containers and are of three types - `Arrays`, `DenseAxisArrays`, and `SparseAxisArrays`.

1.2.1 Arrays

JuMP arrays are created in a similar syntax to Julia arrays with the addition of specifying that the indices start with 1. If we do not tell JuMP that the indices start at 1, it will create a `DenseAxisArray` instead.

```
@variable(model, a[1:2, 1:2])
```

```
2×2 Array{JuMP.VariableRef,2}:  
 a[1,1]  a[1,2]  
 a[2,1]  a[2,2]
```

An n -dimensional variable $x \in R^n$ having a bound $l \preceq x \preceq u$ ($l, u \in R^n$) is added in the following manner.

```
n = 10  
l = [1; 2; 3; 4; 5; 6; 7; 8; 9; 10]  
u = [10; 11; 12; 13; 14; 15; 16; 17; 18; 19]
```

```
@variable(model, l[i] <= x[i=1:n] <= u[i])
```

```
10-element Array{JuMP.VariableRef,1}:  
 x[1]  
 x[2]  
 x[3]  
 x[4]  
 x[5]  
 x[6]  
 x[7]  
 x[8]  
 x[9]  
 x[10]
```

Note that while working with Containers, we can also create variable bounds depending upon the indices

```
@variable(model, y[i=1:2, j=1:2] >= 2i + j)
```

```
2×2 Array{JuMP.VariableRef,2}:  
 y[1,1]  y[1,2]  
 y[2,1]  y[2,2]
```

1.2.2 DenseAxisArrays

`DenseAxisArrays` are used when the required indices are not one-based integer ranges. The syntax is similar except with an arbitrary vector as an index as opposed to a one-based range.

An example where the indices are integers but do not start with one.

```
@variable(model, z[i=2:3, j=1:2:3] >= 0)
```

2-dimensional DenseAxisArray{JuMP.VariableRef,2,...} with index sets:

Dimension 1, 2:3

Dimension 2, 1:2:3

And data, a 2×2 Array{JuMP.VariableRef,2}:

z[2,1] z[2,3]

z[3,1] z[3,3]

Another example where the indices are an arbitrary vector.

```
@variable(model, w[1:5, ["red", "blue"]] <= 1)
```

2-dimensional DenseAxisArray{JuMP.VariableRef,2,...} with index sets:

Dimension 1, 1:5

Dimension 2, ["red", "blue"]

And data, a 5×2 Array{JuMP.VariableRef,2}:

w[1,red] w[1,blue]

w[2,red] w[2,blue]

w[3,red] w[3,blue]

w[4,red] w[4,blue]

w[5,red] w[5,blue]

1.2.3 SparseAxisArrays

SparseAxisArrays are created when the indices do not form a rectangular set. For example, this applies when indices have a dependence upon previous indices (called triangular indexing).

```
@variable(model, u[i=1:3, j=i:5])
```

JuMP.Containers.SparseAxisArray{JuMP.VariableRef,2,Tuple{Any,Any}} with 12 entries:

[2, 3] = u[2,3]

[2, 2] = u[2,2]

[2, 5] = u[2,5]

[1, 4] = u[1,4]

[3, 3] = u[3,3]

[1, 3] = u[1,3]

[2, 4] = u[2,4]

[1, 1] = u[1,1]

[1, 2] = u[1,2]

[1, 5] = u[1,5]

[3, 4] = u[3,4]

[3, 5] = u[3,5]

We can also conditionally create variables by adding a comparison check that depends upon the named indices and is separated from the indices by a semi-colon (;).

```
@variable(model, v[i=1:9; mod(i, 3)==0])
```

JuMP.Containers.SparseAxisArray{JuMP.VariableRef,1,Tuple{Any}} with 3 entries:

[9] = v[9]

[3] = v[3]

[6] = v[6]

1.3 Variable Types

The last argument to the `@variable` macro is usually the variable type. Here we'll look at how to specify the variable type.

1.3.1 Integer Variables

Integer optimization variables are constrained to the set $x \in \mathbb{Z}$

```
@variable(model, intx, Int)
```

or

```
@variable(model, intx, integer=true)
```

intx

1.3.2 Binary Variables

Binary optimization variables are constrained to the set $x \in \{0, 1\}$.

```
@variable(model, binx, Bin)
```

or

```
@variable(model, binx, binary=true)
```

binx

1.3.3 Semidefinite variables

JuMP also supports modeling with semidefinite variables. A square symmetric matrix X is positive semidefinite if all eigenvalues are nonnegative.

```
@variable(model, psdx[1:2, 1:2], PSD)
```

```
2×2 LinearAlgebra.Symmetric{JuMP.VariableRef,Array{JuMP.VariableRef,2}}:
 psdx[1,1]  psdx[1,2]
 psdx[1,2]  psdx[2,2]
```

We can also impose a weaker constraint that the square matrix is only symmetric (instead of positive semidefinite) as follows:

```
@variable(model, symx[1:2, 1:2], Symmetric)
```

```
2×2 LinearAlgebra.Symmetric{JuMP.VariableRef,Array{JuMP.VariableRef,2}}:
 symx[1,1]  symx[1,2]
 symx[1,2]  symx[2,2]
```