

Integer Programming

Arpit Bhatia

June 24, 2019

While we already know how to set a variable as integer or binary in the `@variable` macro, this tutorial covers other JuMP features for integer programming along with some modelling techniques.

`using JuMP`

1 Modelling Logical Conditions

Generally, in a mathematical programming problem, all constraints must hold. However, we might want to have conditions where we have some logical conditions between constraints. In such cases, we can use binary variables for modelling logical conditions between constraints.

1.1 Disjunctive Constraints (OR)

Suppose that we are given two constraints $a'x \geq b$ and $c'x \geq d$, in which all components of a and c are non-negative. We would like to model a requirement that at least one of the two constraints is satisfied. For this, we defined a binary variable y and impose the constraints:

$$\begin{aligned}a'x &\geq yb \\ c'x &\geq (1 - y)d \\ y &\in \{0, 1\}\end{aligned}$$

```
a = rand(1:100,5,5)
c = rand(1:100,5,5)
b = rand(1:100,5)
d = rand(1:100,5)
```

```
model = Model()
@variable(model, x[1:5])
@variable(model, y, Bin)
@constraint(model, a * x .>= y .* b)
@constraint(model, c * x .>= (1 - y) .* d)
```

```
5-element Array{JuMP.ConstraintRef{JuMP.Model,MathOptInterface.ConstraintIndex{MathOptInterface.ScalarAffineFunction{Float64},MathOptInterface.GreaterThan{Float64}},JuMP.ScalarShape},1}:

```

```

44 x[1] + 12 x[2] + 25 x[3] + 60 x[4] + 72 x[5] + 83 y ≥ 83.0
70 x[1] + x[2] + 91 x[3] + 51 x[4] + 100 x[5] + 95 y ≥ 95.0
18 x[1] + 63 x[2] + 66 x[3] + 67 x[4] + 35 x[5] + 56 y ≥ 56.0
46 x[1] + 30 x[2] + 16 x[3] + 44 x[4] + 16 x[5] + 57 y ≥ 57.0
67 x[1] + 6 x[2] + 55 x[3] + 28 x[4] + 38 x[5] + 7 y ≥ 7.0

```

1.2 Conditional Constraints (\implies)

Suppose we want to model that a certain linear inequality must be satisfied when some other event occurs. i.e. for a binary variable z , we want to model the implication

$$z = 1 \implies a^T x \leq b$$

If we know in advance an upper bound $a^T x \leq b$. Then we can write the above as a linear inequality

$$a^T x \leq b + M(1 - z)$$

```

a = rand(1:100,5,5)
b = rand(1:100,5)
m = rand(10000:11000,5)

model = Model()
@variable(model, x[1:5])
@variable(model, z, Bin)
@constraint(model, a * x .<= b .+ (m .* (1 - z)))
# If z was a regular Julia variable, we would not have had to use the vectorized dot
operator

```

2 Modelling Integer Variables

2.1 Integer Variables using Constraints

We can add binary and integer restrictions to the domain of each variable using the `@constraint` macro as well.

```

model = Model()

@variable(model, x)
@variable(model, y)
@constraint(model, x in MOI.ZeroOne())
@constraint(model, y in MOI.Integer())

```

yinteger

2.2 Semi-Continuous Variables

A semi-continuous variable is a continuous variable between bounds $[l, u]$ that also can assume the value zero. ie.

$$x \in \{0\} \cup \{l, u\}$$

```
lower_bound = 7.45
upper_bound = 22.22
@variable(model, a)
@constraint(model, a in MOI.Semicontinuous(lower_bound, upper_bound))
```

$a \in \text{MathOptInterface.SemicontinuousFloat64}(7.45, 22.22)$

2.3 Semi-Integer Variables

A semi-integer variable is a variable which assumes integer values between bounds $[l, u]$ and can also assume the value zero. ie.

$$x \in \{0\} \cup (\{l, u\} \cap \mathbb{Z})$$

```
lower_bound = 5
upper_bound = 34
@variable(model, b)
@constraint(model, b in MOI.Semiinteger(lower_bound, upper_bound))
```

$b \in \text{MathOptInterface.SemiintegerInt64}(5, 34)$

Note that the bounds specified in `MOI.Semiinteger` must be integral otherwise it would throw an error.

3 Special Ordered Sets

3.1 Special Ordered Sets of type 1 (SOS1 or S1)

A Special Ordered Set of type 1 is a set of variables, at most one of which can take a non-zero value, all others being at 0. They most frequently apply where a set of variables are actually 0-1 variables: in other words, we have to choose at most one from a set of possibilities.

```
@variable(model, u[1:3])
@constraint(model, u in MOI.SOS1([1.0, 2.0, 3.0]))
```

$[u_1, u_2, u_3] \in \text{MathOptInterface.SOS1Float64}([1.0, 2.0, 3.0])$

Note that we have to pass `MOI.SOS1` a weight vector which is essentially an ordering on the variables. If the decision variables are related and have a physical ordering, then the weight vector, although not used directly in the constraint, can help the solver make a better decision in the solution process.

3.2 Special Ordered Sets of type 2 (SOS2 or S2)

A Special Ordered Set of type 2 is a set of non-negative variables, of which at most two can be non-zero, and if two are non-zero these must be consecutive in their ordering.

```
@variable(model, v[1:3])  
@constraint(model, v in MOI.SOS2([3.0, 1.0, 2.0]))
```

$[v_1, v_2, v_3] \in \text{MathOptInterface.SOS2Float64}([3.0, 1.0, 2.0])$

The ordering provided by the weight vector is more important in this case as the variables need to be consecutive according to the ordering. For example, in the above constraint, the possible pairs are:

- Consecutive
 - ($x[1]$ and $x[3]$) as they correspond to 3 and 2 resp. and thus can be non-zero
 - ($x[2]$ and $x[3]$) as they correspond to 1 and 2 resp. and thus can be non-zero
- Non-consecutive
 - ($x[1]$ and $x[2]$) as they correspond to 3 and 1 resp. and thus cannot be non-zero